Call32 Interface for FoxPro 2.x

by Rick Strahl

This document describes a method I devised to call Win32 API functions when running FoxPro 2.x under Windows 95 or Windows NT. The idea is based on an existing DLL named CALL32 by Peter Golde, who wrote the this public domain library to interface Visual Basic to the WIN32 API. I added a few custom functions to this library and am providing a handful of custom FoxPro procedures that make calling this interface a bit easier.

Note, that there are a few limitations that are imposed by FoxPro. As with standard API function calls through Foxtools you can't access complex data structures in most cases. In addition, the Call32/WIN32 API use LONG integers - Foxtools has problems with some large integer values and passing them through to the API or other 32 bit DLLs.

What's contained in the Call32.dll:

Declare32()

The function that is used to register 32 bit DLL calls much like you register standard Foxtools functions.

long _export PASCAL Declare32 (

LPSTR lpstrName,       // function name

LPSTR lpstrLib,        // function library

LPSTR lpstrArg         // string indicating arg types )

Call32()

An alias name used to call the actual WIN32 or 32 bit DLL function with the custom  parameters. This function sets up a pointer interface to allow you to pass your parameters declared in Declare 32.

void _export PASCAL Call32(long iProc)

W32Version

Returns the *correct* Windows Version number in 32 bit environments. The value is returned as multiplied by 100. So 3.51 returns 351 and 4.00 returns 400.

int PASCAL _export W32Version()

RegOpen()

Wrapper function for the RegOpenKey API call to overcome problems with the the large LONG key values required. FoxPro chokes on some of the HKEY keys and this function provides a workaround by allowing the HKEY values to be passed as strings. Once you've received a key handle you can use the Call32 functions to access the other registry functions directly (or using the Fox wrappers provided in Call32.prg).

long PASCAL _export RegOpen(LPCSTR cHKey, LPSTR Subkey)

ReadRegistry()

I added this function to provide easy read access to the registry.

long PASCAL _export ReadRegistry

(LPCSTR cHKey,LPSTR cSubkey,LPSTR Value,LPSTR Result,WORD ResultLength)

HiWord()

Returns the High Word of a Double Word integer

int FAR PASCAL _export HiWord(DWORD value)

LoWord()

Returns the Low Word of a Double Word integer

int FAR PASCAL _export LoWord(DWORD value)

HiByte()

Returns the High Byte of a Word

int FAR PASCAL _export HiByte(WORD value)

LoByte()

Returns the Low Byte of a Word

int FAR PASCAL _export LoByte(WORD value)

FoxPro Library Routines:

The following information is probably more important to you, than the actual DLL functions. In order to simplify access from your FoxPro application I've created several wrapper functions for accessing 32 bit DLLs, as well as providing front ends for registry access and retrieving the Windows Version number.

The Call32 Interface:

The Call32 interface involves calling 32 bit API functions through an intermediary DLL interface which results in double declaring the API calls. This can be somewhat confusing and for this reason I created some front end routines that break the process down into 3 logical steps:

· Register the 32 bit DLL function with the Call32 DLL using the Reg32 function.

· Register the 32 bit DLL function with FoxPro's RegFn() using the RegFP function.

· Call the actual function using standard CallFn() syntax.

Here's a simple example (for more examples see the DEMO.PRG file):

   *** CALL WIN32 SetFileAttributes

   ***  BOOL SetFileAttributes(lpFileName, dwFileAttributes)

   ***

   ***  LPCTSTR lpFileName;
/* address of filename
*/

   ***  DWORD dwFileAttributes;
/* address of attributes to set
*/

   #DEFINE file_readonly
1

   #DEFINE file_hidden  
2

   #DEFINE file_system

4

   #DEFINE file_normal

128

   *** Register 32 bit function with the Call32 DLL interface

   lhcall32=reg32("SetFileAttributes","Kernel32.dll","pi")

   *** get a handle for use with Foxtools

   lhsetattr=RegFP("CL","L")

   *** Now actually call the 32 bit function and make file read only

   =callfn(lhsetattr,"Test.txt",file_readonly,lhcall32)

The call to Reg32 starts off the process. This function registers the 32 bit DLL function with the Call32 DLL. The syntax is:

   32bitHandle=Reg32(<cDllFunction>,<cDLLname>,<cParameterTypes>)

Note that the parameter types are different from the RegFn()/CallFn() interface normally used by FoxTools for DLL access. The parameter types, like Foxtools, are passed as individual characters describing the type. But the types are different. There are only three types:

    I -   Any integer value. Note that these will be LONGs in FoxPro since

          a 32 bit integer translates to a 16 bit long.

    P -   Pointer. This includes all strings and any values passed by reference.

    W -   Window and other handles that need to be translated from 16 to 32 bits.

          In essence use this parameter type on any Windows handles like Hwnds, Module Handles etc.

          You might have to experiment with this type and I to get the proper type.

The RegFP function handles setting up the RegFn() call for the 32 bit DLL function. This wrapper is used to hide the fact that you always have to include the handle of the 32 bit function returned by Reg32 as the last parameter of the function. RegFP automatically registers this last parameter.

The syntax uses FoxTools conventions for registering the DLL function:

    FPHandle=RegFP(<cParmTypes>,<cReturnTypes>)

Note that any 32 API integers must be passed as LONGs, since 32 bit integers are 32 bit values.

Finally you can call the DLL function directly with CallFn(). There's one slight difference to a standard CallFN() call in that the last parameter of the call must be the 32 bit function handle that was retrieved with the RegFP() function. You are in essence passing two function pointers one for the aliased function in the 16 bit address space and one for the actual API function. The syntax looks like this:

  lvResult=CallFn(FPHandle,<parm1..ParmN>,32bitHandle)

W32Version

Returns the Windows Version number multiplied by 100.

lnVersion=W32Version()

REGISTRY FUNCTIONS:

RdRegStr

Reads a string value from the registry with a single function call.

lcValue=RdRegStr(cHKEY,cSubKey,cKey,pnLength)

     Pass: cRoot   -   string of registry key values

                        "HKEY_CLASSES_ROOT"

                        "HKEY_CURRENT_USER"

                        "HKEY_LOCAL_MACHINE"

                        "HKEY_USERS"

           cSubKey -   Subkey 'path' ("SOFTWARE\Windows")

           cValue  -   Actual key to read

           nLength -   Buffer Length (optional - 2048*)

Example:

      lcroot="HKEY_LOCAL_MACHINE"

      lcsubkey="SOFTWARE\Microsoft\Windows\CurrentVersion"

      lcvalue="VersionNumber"

      ? "Windows 95 Version from Registry:",rdregstr(lcroot,lcsubkey,lcvalue)

RdRegInt

Identical to RdRegStr except it reads an integer value.

**********************

The following functions are lower level registry routines to allow more control when accessing the registry.

**********************

OpenKey

Opens registry key.

Syntax:

 nHandle=OpenKey(cHKEY,cSubKey)

Example:

*** Read Write Demo

lcroot="HKEY_LOCAL_MACHINE"

lcsubkey="SOFTWARE\Microsoft\Windows\CurrentVersion"

*** Must open the key first

lhreg=openkey(lcroot,lcsubkey)

CloseKey

Closes an open registry based on an open registry handle.

Syntax:

  =CloseKey(nHandle)

QueryStr

--------

Reads a character key value from the registry.

Syntax:

  cValue=QueryStr(nHandle,cEntry,lBinary)

Example:

lcOldSetting=querystr(lhreg,"RegisteredOwner")

QueryInt

Same as QueryStr except it reads an integer.

WriteStr

Writes a character key value to the registry.

Syntax:

  lResult=WriteStr(nHandle,cEntry,cNewValue) 

Example:

  IF !writestr(lhreg,"RegisteredOwner","New Owner")

     wait window "Error writing registry..."

     RETURN

  ENDIF 

WriteInt

Same as WriteStr except it writes an integer value.   

***********************************************************************

PROCEDURE reg32

******************

***    Author: Rick Strahl

***            (c) West Wind Technologies, 1995

***   Contact: (503) 386-2087  / 76427,2363@compuserve.com

***  Modified: 08/22/95

***  Function: Registers a 32 bit DLL function using CALL32.DLL

***            Thunk interface

***    Assume: Foxtools is loaded. CALL32.DLL is available

***      Pass: pcDLLFunction  -   Name of 32 bit function

***            pcDLLName      -   The DLL it is contained in

***            pcParmTypes    -   Parameter types

***                               types: "i"   - Integer (long)

***                                      "w"   - Window Handles

***                                      "p"   - Pointers (all strings)

***

***    Return: Function handle that must be used to CALL32 function

***

***    Partial Docs for Call32:

***

***    1. Summary

***    ----------

***

***    CALL32.DLL is a DLL that can be used for calling routines in 32-bit

***    DLLs on Windows NT/95.  Using it, a Visual Basic program, running

***    in the Win16 subsystem, can declare and call functions in any 32-bit

***    DLL (including, but not limited to, the system DLLs).  CALL32.DLL works

***    on both the x86 and MIPS versions on NT.  It has not been tested

***    on Alpha or other versions, but should work.

***

***

***    2. Usage

***    --------

***

***    To call a function in a 32-bit DLL, follow the following steps.  Declare

***    the "Declare32" function as follows (all one one line):

***

***    Declare Function Declare32 Lib "call32.dll" (ByVal Func$,

***           ByVal Library$, ByVal Args$) As Long

***

***    Next, declare the function you wish to call.  Declare it in the ordinary

***    fashion, with the following exceptions:

***       - Use a library name of "call32.dll"

***       - Use an Alias of "Call32"

***       - Add an additional argument at the end, of type ByVal Long

***    For example, if you are calling the function:

***      GetWindowText(HWND hwnd, LPSTR lpsz, int cch)

***    declare it as follows (remember that ints and all handles are 32 bits,

***    so use a Long):

***      Declare Function GetWindowText Lib "call32.dll" Alias "Call32"

***            (ByVal hwnd As Long, ByVal lpsz As String,

***             ByVal cch As Long, ByVal id As Long) As Long

***

***    In the initialization section of your application, you declare the

***    actual library and name of the function you want to call with

***    the Declare32 function.  Pass it the name of the function, the

***    library, and a string describing the argument types.  Each letter

***    in the string declares the type of one argument, and should be

***    either "i" for a 32 bit integer or handle type, "p" for any

***    pointer type, or "w" for an HWND parameter you want to pass

***    a 16 bit HWND to and have be automatically converted to a 32 bit

***    HWND.  The return value of Declare32 should be saved away in

***    a global variable to be passed as the last parameter to the

***    function you declared earlier.  So, continue the example, you

***    would call:

***

***      idGetWindowText = Declare32("GetWindowText", "user32", "wpi")

***

***    (As a side note, this would be more properly declared as

***    "GetWindowTextA", since this is the real exported name.  However,

***    Declare32 will automatically add an "A" to the end of a

***    function name if necessary).

***

***    To call the function, you would just call:

***

***      cbCopy = GetWindowText(hwnd, sz, cb, idGetWindowText)

***

***  FoxPro Example:

***     lnCall32=Reg32("GetVersion","Kernel32.dll","")

***

***     *** Alternately, this is the manual approach:

***     *** lhCall32=regfn("Declare32","CCC","L","CALL32.DLL")

***     *** lnCall32=callfn(lhCall32,"GetVersion","Kernel32.DLL","")

***     *** ? lnCall32

***

***     *** Note the extra Long parameter even though no parm

***     *** is required by the API function!

***     *** You must always call CALL32!!! The long parm

***     *** passed last actually routes the call to the

***     *** proper 32 bit function!

***

***

***     lhWinVersion=RegFP("","L")

***     *** Alternately you can use this manual approach:

***     *** lhWinVersion=regfn("Call32","L","L","CALL32.DLL")

***     lnVersion=callfn(lhWinVersion,lnCall32)

***

************************************************************************

PARAMETERS pcdllfunction,pcdllname,pcparmtypes

PRIVATE lhcall32,lncall32

lhcall32=regfn("Declare32","CCC","L","CALL32.DLL")

lncall32=callfn(lhcall32,pcdllfunction,pcdllname,pcparmtypes)

RETURN lncall32

************************************************************************

PROCEDURE regfp

******************

***    Author: Rick Strahl

***            (c) West Wind Technologies, 1995

***   Contact: (503) 386-2087  / 76427,2363@compuserve.com

***  Modified: 08/22/95

***  Function: Wrapper for Call32 function, allows direct call

***            of 32 bit DLLs passing parameters as String!

***    Assume: FoxTools loaded, CALL32.DLL available

***      Pass: pcParms      -   Parameter types (like RegFn())

***            pcRetType    -   Return Type (like RegFn()))

***    Return: Function Return value

************************************************************************

PARAMETER pcparms,pcrettype

RETURN regfn("Call32",pcparms+"L",pcrettype,"CALL32.DLL")

************************************************************************

PROCEDURE W32Version

******************

***  Modified: 01/06/96

***  Function: Returns Windows Version Number (multiplied by 100)

*************************************************************************

lhw32ver=regfn("W32Version","","I","call32.dll")

RETURN callfn(lhw32ver)

************************************************************************

PROCEDURE rdregstr

******************

***    Author: Rick Strahl

***            (c) West Wind Technologies, 1995

***   Contact: (503) 386-2087  / 76427,2363@compuserve.com

***  Modified: 08/24/95

***  Function: Reads a single String value from the Registry

***    Assume: Requires Call32.dll

***            Probably works on Binary entries as well

***      Pass: pcRoot  -   string of registry key values

***                        "HKEY_CLASSES_ROOT"

***                        "HKEY_CURRENT_USER"

***                        "HKEY_LOCAL_MACHINE"

***                        "HKEY_USERS"

***            pcSubKey-   Subkey 'path' ("SOFTWARE\Windows")

***            pcValue -   Actual key to read

***    Return: key value, "", "*ERROR*"

************************************************************************

PARAMETERS pcroot,pcsubkey,pckey,pnlength

PRIVATE lcresult,lnlength,lnresult

IF PARAMETERS()<3 OR ;



TYPE("pcRoot")#"C" OR ;



TYPE("pcSubKey")#"C" OR ;



TYPE("pcKey")#"C"


RETURN "*ERROR*"

ENDIF

pnlength=IIF(TYPE("pnLength")="N",pnlength,2048)

lcresult=SPACE(pnlength)

*** long PASCAL _export ReadRegistry

*** (LPCSTR cHKey,LPSTR Subkey,LPSTR Value,LPSTR Result,INT Length)

lhreg=regfn("ReadRegistry","C@C@C@CI","L","call32.dll")

lnresult=callfn(lhreg,lcroot,@pcsubkey,@pckey,@lcresult,pnlength)

*** Return only left of Null

lcresult=TRIM(LEFT(lcresult,AT(CHR(0),lcresult)-1))

IF lnresult#0


RETURN "*ERROR*"

ENDIF

RETURN lcresult

*EOP RegString

************************************************************************

PROCEDURE rdregint

******************

***    Author: Rick Strahl

***            (c) West Wind Technologies, 1995

***   Contact: (503) 386-2087  / 76427,2363@compuserve.com

***  Modified: 08/24/95

***  Function: Reads a single String value from the Registry

***    Assume: Requires Call32.dll

***      Pass: pcRoot  -   string of registry key values

***                        "HKEY_CLASSES_ROOT"

***                        "HKEY_CURRENT_USER"

***                        "HKEY_LOCAL_MACHINE"

***                        "HKEY_USERS"

***            pcSubKey-   Subkey 'path' ("SOFTWARE\Windows")

***            pcValue -   Actual key to read

***            pnLength-   Byte length (16)

***    Return: key value or -1 0 if empty or non-existant

************************************************************************

PARAMETERS pcroot,pcsubkey,pckey,pnlength

PRIVATE lcresult,lnlength,lnresult

IF PARAMETERS()<3 OR ;



TYPE("pcRoot")#"C" OR ;



TYPE("pcSubKey")#"C" OR ;



TYPE("pcKey")#"C"


RETURN -1

ENDIF

pnlength=IIF(TYPE("pnLength")="N",pnlength,16)

lnresult=0

lnlength=16

*** long PASCAL _export ReadRegistry

*** (LPCSTR cHKey,LPCSTR Subkey,LPCSTR Key,LPSTR Result,DWORD Length)

*** Note here the pointer to receive the value is changed to a Long

*** in the RegFn() declaration.

lhreg=regfn("ReadRegistry","C@C@C@LI","L","call32.dll")

lnretval=callfn(lhreg,lcroot,@pcsubkey,@pckey,@lnresult,pnlength)

IF lnretval#0


RETURN -1

ENDIF

RETURN lnresult

*EOP RdRegInt

************************************************************************

PROCEDURE openkey

******************

***  Modified: 12/08/95

***  Function: Opens a registry key before reading

***            writing entries.

***    Assume: Calls 16 bit Regopen in CALL32.DLL

***             because of limitations in Longs & HKEY

***      Pass: tcHkey   -  "HKEY_..." strings

***            tcSubkey -  Reg path "\Software\version"

***    Return: key handle

*************************************************************************

PARAMETER lchkey,lcsubkey

lnrethandle=0

lhropen=regfn("RegOpen","CC","L","CALL32.DLL")

lnkey=callfn(lhropen,lchkey,lcsubkey)

RETURN lnkey

************************************************************************

PROCEDURE closekey

******************

***  Function: Close registry key.

***      Pass: thHandle - Key handle

*************************************************************************

PARAMETER thhandle

lhclose=regfn("RegCloseKey","L","L","shell.dll")

RETURN callfn(lhclose,thhandle)

************************************************************************

FUNCTION querystr

******************

***  Modified: 12/07/95

***  Function: Reads a registry string

***     Notes: Also works with Binary types

***            as long as it doesn't contain

***            NULL values.

***      Pass: tnHandle - Key Handle

***            tcEntry  - Entry to retrieve

***    Return: string or "" if empty or "*ERROR*"

************************************************************************

PARAMETER tnhandle,tcentry,tlbinary

PRIVATE lhcall32,lhfp,lcdatabuffer

lhcall32=reg32("RegQueryValueEx","ADVAPI32.dll","ipippp")

lhfp=regfp("LCL@L@C@L","L")

*** Return buffer to receive value

lcdatabuffer=SPACE(1024)

lnsize=LEN(lcdatabuffer)

lntype=1         && REG_SZ

lnresult=callfn(lhfp,tnhandle,tcentry,0,@lntype,;


@lcdatabuffer,@lnsize,lhcall32)

IF lnresult#0


RETURN "*ERROR*"

ENDIF

IF lnsize<2


RETURN ""

ENDIF

*** Return string and strip out NULLs

RETURN SUBSTR(CHRTRAN(lcdatabuffer,CHR(0),""),1,lnsize-1)

* ReadRegistryString

************************************************************************

FUNCTION queryint

******************

***    Author: Rick Strahl

***            (c) West Wind Technologies, 1995

***   Contact: (503) 386-2087  / 76427.2363@compuserve.com

***  Modified: 12/07/95

***  Function: Reads a registry string

***      Pass: tnHandle - Key Handle

***            tcEntry  - Entry to retrieve

***    Return: string or "" if empty or "*ERROR*"

************************************************************************

PARAMETER tnhandle,tcentry

PRIVATE lhcall32,lhfp,lcdatabuffer

lhcall32=reg32("RegQueryValueEx","ADVAPI32.dll","ipippp")

lhfp=regfp("LCL@L@L@L","L")

*** Return buffer to receive value

lndatabuffer=0

lnsize=8

lntype=1         && REG_SZ

lnresult=callfn(lhfp,tnhandle,tcentry,0,@lntype,;


@lndatabuffer,@lnsize,lhcall32)

IF lnresult#0


RETURN -1

ENDIF

RETURN lndatabuffer

* ReadRegistryString

************************************************************************

PROCEDURE writestr

******************

***  Modified: 12/08/95

***  Function: Writes a key value to an entry.

***    Assume: Uses Call32.dll

***      Pass: tnHandle - Key Handle

***            tcEntry  - The entry off the key

***            tcValue  - Value to set it to

***    Return: .T. or .F.

*************************************************************************

PARAMETERS tnhandle, tcentry, tcvalue

PRIVATE  lnsize, lcdatabuffer, tntype

lhreg32=reg32("RegSetValueEx","advapi32.dll","ipiipi")

lhregfp=regfp("LCLLCL","L")

lnsize=LEN(tcvalue)

lntype=1   && REG_SZ

lnresult=callfn(lhregfp,tnhandle,tcentry,0,1,;


tcvalue,lnsize,lhreg32)

IF lnresult#0


RETURN .F.

ENDIF

RETURN .T.

* WriteStr

************************************************************************

PROCEDURE writeint

******************

***  Modified: 12/08/95

***  Function: Writes a key value to an entry.

***    Assume: Uses Call32.dll

***      Pass: tnHandle - Key Handle

***            tcEntry  - The entry off the key

***            tnValue  - Value to set it to

***    Return: .T. or .F.

*************************************************************************

PARAMETERS tnhandle, tcntry, tnvalue

PRIVATE  lnsize, lcdatabuffer, lntype

lhreg32=reg32("RegSetValueEx","advapi32.dll","ipiiii")

lhregfp=regfp("LCLLLL","L")

lnsize=8

lntype=4   && REG_DWORD

lnresult=callfn(lhregfp,tnhandle,tcentry,0,lntype,;


tnvalue,lnsize,lhreg32)

IF lnresult#0


RETURN -1

ENDIF

RETURN tnvalue

* WriteStr

